/****************************************************************************
**
** Copyright (C) 2019 Luxoft Sweden AB
** Copyright (C) 2018 Pelagicore AG
** Contact: https://www.qt.io/licensing/
**
** This file is part of the Neptune 3 UI.
**
** $QT_BEGIN_LICENSE:GPL-QTAS$
** Commercial License Usage
** Licensees holding valid commercial Qt Automotive Suite licenses may use
** this file in accordance with the commercial license agreement provided
** with the Software or, alternatively, in accordance with the terms
** contained in a written agreement between you and The Qt Company.  For
** licensing terms and conditions see https://www.qt.io/terms-conditions.
** For further information use the contact form at https://www.qt.io/contact-us.
**
** GNU General Public License Usage
** Alternatively, this file may be used under the terms of the GNU
** General Public License version 3 or (at your option) any later version
** approved by the KDE Free Qt Foundation. The licenses are as published by
** the Free Software Foundation and appearing in the file LICENSE.GPL3
** included in the packaging of this file. Please review the following
** information to ensure the GNU General Public License requirements will
** be met: https://www.gnu.org/licenses/gpl-3.0.html.
**
** $QT_END_LICENSE$
**
** SPDX-License-Identifier: GPL-3.0
**
****************************************************************************/

import QtQuick 2.10
import centerconsole 1.0
import shared.animations 1.0
import system.models.notification 1.0
import shared.Sizes 1.0

ModalOverlay {
    id: root

    showModalOverlay: notificationCenter.openNotificationCenter
    onOverlayClicked: {
        notificationCenter.closeNotificationCenter();
    }

    NotificationModel {
        id: notificationModel
        onNotificationsCleared: {
            notificationHandle.forceActiveFocus();
        }
        onNotificationClosed: {
            notificationHandle.forceActiveFocus();
        }
    }

    NotificationCenter {
        id: notificationCenter
        anchors.left: parent.left
        anchors.right: parent.right
        notificationModel: notificationModel
    }

    NotificationHandle {
        id: notificationHandle
        anchors.horizontalCenter: root.horizontalCenter
        anchors.top: notificationCenter.bottom
        notificationCount: notificationModel.count
        notificationCounterVisible: ((notificationCount > 0) && !notificationCenter.openNotificationCenter)
        dragTarget: notificationCenter
        minimumY: -notificationCenter.height
        maximumY: 0
        onActiveChanged: {
            if (active) {
                notificationHandle.prevDragY = notificationHandle.dragTarget.y;
            } else {
                if (!notificationHandle.dragActive && notificationHandle.swipe) {
                    if (notificationCenter.openNotificationCenter && notificationHandle.dragDelta > 0) {
                        notificationCenter.openNotificationCenter = true;
                    } else if (notificationCenter.openNotificationCenter && notificationHandle.dragDelta < 0) {
                        notificationCenter.openNotificationCenter = false;
                    } else {
                        notificationCenter.openNotificationCenter = !notificationCenter.openNotificationCenter;
                    }
                } else {
                    notificationCenter.openNotificationCenter = !notificationCenter.openNotificationCenter;
                }

                // stop drag filter timer
                notificationHandle.dragFilterTimer.running = false;
                notificationHandle.dragDelta = (notificationCenter.y - notificationHandle.dragOrigin + Sizes.dp(130));
            }
        }

        onClicked: {
            //open notificationCenter
            notificationCenter.openNotificationCenter = !notificationCenter.openNotificationCenter;
        }

        onPressed: {
            // reset values
            notificationHandle.dragDelta = 0;
            notificationHandle.dragOrigin = notificationHandle.dragTarget.y;
            notificationHandle.prevDragY = notificationHandle.dragTarget.y;

            // start drag filter timer
            notificationHandle.dragFilterTimer.running = true;
        }
    }
}
